/*-----------------------------------------------------------------------------
	The contents of this file are subject to the Mozilla Public License
	Version 1.1 (the "License"); you may not use this file except in
	compliance with the License. You may obtain a copy of the License at
	http://www.mozilla.org/MPL/
	
	Software distributed under the License is distributed on an "AS IS"
	basis, WITHOUT WARRANTY OF ANY KIND, either express or implied. See the
	License for the specific language governing rights and limitations
	under the License.
	
	The Original Code is Fever Framework code.
	
	The Initial Developer of the Original Code is Romain Ecarnot.
	Portions created by Initial Developer are Copyright (C) 2006
	the Initial Developer. All Rights Reserved.
	
	Contributor(s):
	
	Unless required by applicable law or agreed to in writing, software
	distributed under the License is distributed on an "AS IS" BASIS,
	WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
	See the License for the specific language governing permissions and
	limitations under the License.
-----------------------------------------------------------------------------*/

import org.aswing.JLabel;

import com.bourre.commands.CommandManagerMS;
import com.bourre.commands.Delegate;

import fever.data.formatter.DateFormatter;
import fever.utils.DateResources;

/**
 * Simple label component with automatic today's date.
 * 
 * <p>Date is formatting using {@link fever.data.formatter.DateFormatter} formatter 
 * and the <strong>Fever Localisation API</strong>.
 *
 * @author Romain Ecarnot
 */
class fvaswing.components.FvDateLabel extends JLabel
{
	//-------------------------------------------------------------------------
	// Private properties
	//-------------------------------------------------------------------------
	
	private var _formatter : DateFormatter;
	private var _clock : Delegate;
		
		
	//-------------------------------------------------------------------------
	// Public API
	//-------------------------------------------------------------------------
	
	/**
	 * Constructor.
	 */
	public function FvDateLabel( HAlignment : Number )
	{
		super( "", HAlignment );
		
		_formatter = new DateFormatter();
		_formatter.formatString = DateResources.getInstance().formatting;
		
		addEventListener( JLabel.ON_CREATED, _create, this );
		addEventListener( JLabel.ON_DESTROY, _release, this );
	}
	
	/**
	 * Sets the formatting pattern for date representation.
	 */
	public function setFormattingString( pattern : String ) : Void
	{
		if( _formatter.formatString != pattern )
		{
			_formatter.formatString = pattern;
			_update();
		}	
	}
	
	/**
	 * Returns the formatting pattern apply.
	 */
	public function getFormattingString() : String
	{
		return _formatter.formatString;	
	}
	
	
	//-------------------------------------------------------------------------
	// Private implementation
	//-------------------------------------------------------------------------
	
	private function _create() : Void
	{
		_clock = new Delegate( this, _update );
		CommandManagerMS.getInstance().push( _clock, 1000 );
	}
	
	private function _update() : Void
	{
		setText( _formatter.format( new Date() ) );
	}
	
	private function _release() : Void
	{
		CommandManagerMS.getInstance().remove( _clock );
	}
}